

## INTRODUCTION

HSPICE is a circuit simulation program. This tutorial is a guide to its use as a *standalone tool* for performing circuit simulation. In this tutorial HSPICE will be used to perform a *transient* analysis of several CMOS inverter models. These models vary in the complexity of the HSPICE models used to describe the behaviour of the NMOS and CMOS transistors composing the inverter circuit.

A simple inverter circuit will be simulated with the HSPICE program and the results will be analyzed with the AvanWaves tool. This is a tool used to display, analyze, and print the results of HSPICE simulations.

### EXAMPLE 1: A simple CMOS inverter.

Consider the basic CMOS inverter circuit shown below in Figure 1.




Figure 1: Simple CMOS Inverter Circuit.

The HSPICE netlist describing this inverter circuit is:

```
Inverter Circuit
.OPTIONS LIST NODE POST
.TRAN 200P 20N
.PRINT TRAN V(IN) V(OUT)
M1 OUT IN VCC VCC PCH L=1U W=20U
M2 OUT IN 0 0 NCH L=1U W=20U
VCC VCC 0 5
VIN IN 0 0 PULSE .2 4.8 2N 1N 1N 5N 20N
CLOAD OUT 0 .75P
.MODEL PCH PMOS LEVEL=1
.MODEL NCH NMOS LEVEL=1
.END
```

The PMOS and NMOS transistors are given the names M1 and M2 respectively. The two lines in the HSPICE netlist:

```
M1 OUT IN VCC VCC PCH L=1U W=20U
M2 OUT IN 0 0 NCH L=1U W=20U
```

describe the connection of the **drain**, **gate**, **source**, and **bulk(substrate)** terminals, the name of the HSPICE model used to describe electrical characteristics of the transistor, and the **length** and **width** of the channel. For example, the transistor labelled M1 in the circuit (and the netlist) has its drain terminal connected to the node labelled OUT, the gate terminal connected to node IN, both the source and substrate terminals are connected to the node labelled VCC. The name of the HSPICE mosfet model used to describe this transistor's electrical parameters is PCH.

The two .MODEL lines tell HSPICE which transistor model to use. In this case, we are using two basic models called PCH and NCH.

## PERFORMING A TRANSIENT ANALYSIS

1. Type the above netlist into a file called *inverter.sp*.
2. Type the following to run HSPICE.

```
hspice inverter.sp > inverter.lis
```

This will run the HSPICE simulator, reading in the commands in the file *inverter.sp* and generating an output file called *inverter.lis*. In addition, several intermediate files will be created which will be used by the AwanWaves tool to display the simulation results.

3. Once the simulation completes, the following message will be printed:

```
>info:      ***** hspice job concluded
```

```
real    0.5
user    0.2
sys     0.1
```

4. Examine the inverter.lis file. It will contain a listing of the voltages at nodes IN and OUT in the following tabular format:

| time       | voltage<br>in | voltage<br>out |
|------------|---------------|----------------|
| 0.         | 200.0000m     | 4.9989         |
| 200.00000p | 200.0000m     | 4.9989         |
| 400.00000p | 200.0000m     | 4.9989         |
| 600.00000p | 200.0000m     | 4.9989         |
| 800.00000p | 200.0000m     | 4.9989         |
| 1.00000n   | 200.0000m     | 4.9989         |
| 1.20000n   | 200.0000m     | 4.9989         |
| 1.40000n   | 200.0000m     | 4.9989         |
| 1.60000n   | 200.0000m     | 4.9989         |
| 1.80000n   | 200.0000m     | 4.9989         |
| 2.00000n   | 200.0000m     | 4.9989         |
| 2.20000n   | 1.1200        | 4.8697         |
| 2.40000n   | 2.0400        | 4.2160         |
| 2.60000n   | 2.9600        | 2.1088         |
| 2.80000n   | 3.8800        | 582.0086m      |
| 3.00000n   | 4.8000        | 1.4790m        |

This was generated by including the line

```
.PRINT TRAN V(IN) V(OUT)
```

in the HSPICE netlist file inverter.sp.

5. To view the simulation results using the Synopsys Spice Viewer <sup>TM</sup>tool, type the following from the UNIX prompt:

```
source /CMC/ENVIRONMENT/hspice.env
```

This will set the user path to point to the location of the 'wv' binary executable. To start the Spice Viewer, type the following from the UNIX prompt:

```
wv &
```

After a few seconds, the Waveform Analyzer window will appear as shown in Figure 2.




Figure 2: Waveview Analyzer window.

6. To read in simulation results such as a .tr0 or an .ac0 file select:

File -> Import Wavefile

The Open: Waveform Files will appear as shown in Figure 3.




Figure 3: Open Waveform window.

7. From the Open: Waveform Files window, select an output file to open and then select **Ok**. The name of the file will appear in the Output View portion of the Waveform Analyzer as indicated in Figure 4.




Figure 4: Output View portion of Waveform window.

8. Double click on this filename and it will expand to show "toplevel". Double click on "toplevel" and a list of nodes from the circuit netlist will appear as shown in Figure 5.




Figure 5: List of nodes in Outview view of Waveform window.

9. To plot a signal, double click on it. It's waveform will appear in the main portion of the Waveform Analyzer. Each double click on a signal name will cause a new panel to appear containing the plot. Refer to Figure 6.




Figure 6: Waveform window with plots in each panel.

10. To quit the Waveform Analyzer select:

File -> Exit

11. The documentation for the Synopsys Spice Viewer is found in:

`/CMC/tools/synopsys/sx/c2009.03/sx_c2009_03/doc/manuals`

A good starting point to learning more about the Spice Viewer tool is the quickref.pdf document (Quick Reference).

### **EXAMPLE 2: CMOS Inverter using different transistor models.**

The previous example used SPICE level 1 models for the nmos and pmos transistors. More accurate models for these devices have been developed. Included in the CMOSIS5 Design Kit for Cadence from the Canadian Microelectronics Corporation are HSPICE models of varying detail. These models are found in the directory `/CMC/kits/cmosis5/cadence/cmosis5.2.2/models/hspice`. This example will show how to use the `cmosis5.level3` HSPICE MOSFET models directly in the HSPICE netlist for the inverter circuit described in the previous model.



1. Copy the file /CMC/kits/cmosis5/cadence/cmosis5.2.3/models/hspice/cmosis5.level3 to your working directory.
2. Copy the inverter.sp file to a new file called cmosis5inverter.sp.
3. Append the contents of the file cmosis5.level3 to the file cmosis5inverter.sp, this can be done using the UNIX cat command:

```
cat cmosis5.level3 > cmosis5inverter.sp
```

4. Using a text editor, edit the cmosis5inverter.sp file so that it is similar to the example netlist given below:.

#### Inverter Circuit

```
.OPTIONS LIST NODE POST
.TRAN 200P 15N
.PRINT TRAN V(IN) V(OUT)
M1 OUT IN VCC VCC CMOSF L=1U W=20U
M2 OUT IN 0 0 CMOSN L=1U W=20U
VCC VCC 0 5
VIN IN 0 0 PULSE .2 4.8 2N 1N 1N 5N 20N
CLOAD OUT 0 .75P

*Run=n5bo
*date=1-Feb-1996

.MODEL CMOSN NMOS LEVEL=3
+ PHI=0.700000 TOX=9.6000E-09 XJ=0.200000U TPG=1
+ VTO=0.6566 DELTA=6.9100E-01 LD=4.7290E-08 KP=1.9647E-04
+ UO=546.2 THETA=2.6840E-01 RSH=3.5120E+01 GAMMA=0.5976
+ NSUB=1.3920E+17 NFS=5.9090E+11 VMAX=2.0080E+05 ETA=3.7180E-02
+ KAPPA=2.8980E-02 CGDO=3.0515E-10 CGSO=3.0515E-10
+ CGBO=4.0239E-10 CJ=5.62E-04 MJ=0.559 CJSW=5.00E-11
+ MJSW=0.521 PB=0.99
+ XW=4.108E-07
* Weff = Wdrawn - Delta_W
* The suggested Delta_W is 4.1080E-07

.MODEL CMOSF PMOS LEVEL=3
+ PHI=0.700000 TOX=9.6000E-09 XJ=0.200000U TPG=-1
+ VTO=-0.9213 DELTA=2.8750E-01 LD=3.5070E-08 KP=4.8740E-05
+ UO=135.5 THETA=1.8070E-01 RSH=1.1000E-01 GAMMA=0.4673
+ NSUB=8.5120E+16 NFS=6.5000E+11 VMAX=2.5420E+05 ETA=2.4500E-02
+ KAPPA=7.9580E+00 CGDO=2.3922E-10 CGSO=2.3922E-10
+ CGBO=3.7579E-10 CJ=9.35E-04 MJ=0.468 CJSW=2.89E-10
+ MJSW=0.505 PB=0.99
+ XW=3.622E-07
* Weff = Wdrawn - Delta_W
* The suggested Delta_W is 3.6220E-07
.END
```

5. Notice that the two lines describing the transistors M1 and M2 have been changed to reflect the new names of the two transistor models. Formerly, the names were PCH and NCH. These have been changed to CMOSP and CMOSN since these are the names given to the models in the .MODEL lines. Also, the two lines .lib n5bo and .endl found at the beginning and end of the transistor model statements have been removed. The reason for this is that we are not using a library file to hold the transistor models; but have included the two models directly in our HSPICE netlist. The use of a library file will be shown in the next two examples.

6. Simulate and view the results using the same commands as used for the previous example.

- (i) hspice cmosis5inverter.sp > cmosis5inverter.lis
- (ii) wv &
- (iii) File -> Import Wavefile -> cmosis5inverter.tr0

### EXAMPLE 3: CMOS Inverter using a library file containing the MOSFET models.

It is not necessary to have the two .MODEL statements containing the MOSFET models contained in the HSPICE netlist. Rather, these statements may be contained in a library file, and the contents of this library file can be read in when performing the simulation by the use of the .LIB statement in the HSPICE netlist. This example illustrates the use of such a library file found in the current working directory. It assumes the existence of a file called cmosi5.level3 in the working directory with the following contents:

```
*Run=n5bo
*date=1-Feb-1996

.lib n5bo
.MODEL CMOSN NMOS LEVEL=3
+ PHI=0.700000 TOX=9.6000E-09 XJ=0.200000U TPG=1
+ VTO=0.6566 DELTA=6.9100E-01 LD=4.7290E-08 KP=1.9647E-04
+ UO=546.2 THETA=2.6840E-01 RSH=3.5120E+01 GAMMA=0.5976
+ NSUB=1.3920E+17 NFS=5.9090E+11 VMAX=2.0080E+05 ETA=3.7180E-02
+ KAPPA=2.8980E-02 CGDO=3.0515E-10 CGSO=3.0515E-10
+ CGBO=4.0239E-10 CJ=5.62E-04 MJ=0.559 CJSW=5.00E-11
+ MJSW=0.521 PB=0.99
+ XW=4.108E-07
* Weff = Wdrawn - Delta_W
* The suggested Delta_W is 4.1080E-07

.MODEL CMOSP PMOS LEVEL=3
+ PHI=0.700000 TOX=9.6000E-09 XJ=0.200000U TPG=-1
+ VTO=-0.9213 DELTA=2.8750E-01 LD=3.5070E-08 KP=4.8740E-05
+ UO=135.5 THETA=1.8070E-01 RSH=1.1000E-01 GAMMA=0.4673
+ NSUB=8.5120E+16 NFS=6.5000E+11 VMAX=2.5420E+05 ETA=2.4500E-02
+ KAPPA=7.9580E+00 CGDO=2.3922E-10 CGSO=2.3922E-10
+ CGBO=3.7579E-10 CJ=9.35E-04 MJ=0.468 CJSW=2.89E-10
+ MJSW=0.505 PB=0.99
+ XW=3.622E-07
* Weff = Wdrawn - Delta_W
```

```
* The suggested Delta_W is 3.6220E-07
.endl
```

1. Create a new file called `cmosis5inverter_with_lib.sp` with the following contents:

```
Inverter Circuit
.LIB 'cmosis5.level3' n5bo
.OPTIONS LIST NODE POST
.TRAN 200P 15N
.PRINT TRAN V(IN) V(OUT)
M1 OUT IN VCC VCC CMOSF L=1U W=20U
M2 OUT IN 0 0 CMOSN L=1U W=20U
VCC VCC 0 5
VIN IN 0 0 PULSE .2 4.8 2N 1N 1N 5N 20N
CLOAD OUT 0 .75P
.END
```

Note the **.LIB 'cmosis5.level3' n5bo** line. This will cause the entry named `n5bo` found in the library file called `'cmosis5.level3'` (note there is no absolute path to this file, hence it is assumed to be found in the present working directory) to be read in. The entry is read until its corresponding `.ENDL` statement is encountered. Libraries are built using the `.LIB` statement in a library file. Commonly used device models, commands, and statements can be placed in a library file. The `.LIB` statement begins the library macro, and the `.ENDL` statement ends the library macro. In our library file, we have only one entry called `n5bo` which contains the HSPICE MOSFET models for the two transistors.

Note how the use of library files results in a much neater HSPICE netlist. Another advantage is that the HSPICE models may be changed by more sophisticated users.

#### **EXAMPLE 4: CMOS Inverter with libraries found in a different location.**

As a final example, one may specify the absolute path to where the CMOSIS5 HSPICE transistor models are found and use these models rather than copying the models to one's working directory. However, using this approach, one cannot modify the files since ordinary users do not have write access to these files. The following HSPICE netlist contains a `.LIB` statement which gives the *absolute path* to where the desired library file is found.

```
Inverter Circuit
.LIB '/CMC/kits/cmosis5/cadence/cmosis5.2.3/models/hspice/cmosis5.level3' n5bo
.OPTIONS LIST NODE POST
.TRAN 200P 15N
.PRINT TRAN V(IN) V(OUT)
M1 OUT IN VCC VCC CMOSF L=1U W=20U
M2 OUT IN 0 0 CMOSN L=1U W=20U
VCC VCC 0 5
VIN IN 0 0 PULSE .2 4.8 2N 1N 1N 5N 20N
CLOAD OUT 0 .75P
.END
```

The only difference between this example and the previous is the .LIB statement.

## PRINTING SIMULATION RESULTS

To print your simulation results, select **File -> Print** from the WaveView main window. The **Print Setup** window will appear. To print to a Postscript file, select **Print to File** and specify a filename in the **Output Path** field. Select **Print** and the Postscript file will be generated. The file may then be printed using the

```
lpr -Pprinter_name filename
```

command.

## OBTAINING MORE INFORMATION ON HSPICE

Electronic documentation in Adobe PDF format is available in the directory:

```
/CMC/tools/meta/docs
```

This directory contains several user guides and quick references for the HSPICE simulator itself. For example, the file hspice\_quickref.pdf found in this directory is the HSPICE Quick Reference Guide. It may be viewed with the command:

```
acroread /CMC/tools/meta/docs/hspice_quickref.pdf
```

Should you care to print it out, be wary you may quickly use up your laser printing quota.